



Rick Hance Engineering Note

Date: 4/8/97
Rev Date: 3/16/98
Project: Solenoid Energization, Controls, Interlocks and Quench Protection
Doc. No: H970408A

Subject: Solenoid Controls Algorithms

The solenoid energization system is controlled and monitored at the I/O level by a programmable logic controller (PLC). The PLC is programmed using Siemens' TISOFT software. Operators and users interface to the PLC via Intellusion's "Fully Integrated Control System" (FIXS), "Distributed Manufacturing and Control" (DMACs) software. This software allows an operator, located at a networked DMACs terminal, to monitor and control the system by influencing the operation of the PLC. The operator's interface is in the form of a hierarchy of pictures. The pictures provide status information and buttons for control. Most control algorithms are merely triggered by DMACs scripts; and actually implemented by ladder logic or special programs in the PLC. The scripts run in response to actions by the operator. Examples of these actions are: selecting a push button on a DMACs picture with the mouse; or the opening of a picture in response to some other action. The following files are the pictures which comprise the control and status interface to the solenoid operator. All pictures are located on the DZERO 'Dmacs-nt-server' drive S. The pictures are located in path S:\WDMACS\PIC\DO*.ODF:

- Dcontrol.odf - Top level solenoid energization system picture. The component blocks on this picture are actually buttons to access status and control information for individual components such as power supply, reversing switch and etc. This top level picture also provides an emergency crash button and access to a series of control's experts diagnostic pages.
- Dc_cs.odf - Current sensing status; provides raw and scaled current values.
- Dc_dr.odf - Dump resistor status.
- Dc_ds.odf - Dump switch status and access to control functions.
- Dc_dsc.odf - Dump switch control for authorized users.
- Dc_emr.odf - Emergency shutoff verification (software crash button).
- Dc_expai.odf - Expert's analog inputs.
- Dc_expao.odf - Expert's analog outputs.
- Dc_expct.odf - Expert's control variables.
- Dc_expcv.odf - Expert's calculated values.
- Dc_expdi.odf - Expert's digital inputs.
- Dc_expdo.odf - Expert's digital outputs.
- Dc_expti.odf - Expert's temperature inputs.
- Dc_gf.odf - Ground fault detector status.
- Dc_in.odf - Interlock status and access to control functions.
- Dc_inc.odf - Interlock control for authorized users.
- Dc_lcw.odf - Low conductivity water cooling & condensation control
- Dc_nf.odf - Noise filter status.
- Dc_ps.odf - Power supply status and access to control functions.
- Dc_psc.odf - Power supply control for authorized users (ON, OFF, RESET, CURRENT).
- Dc_qd.odf - Quench detector status and access to control functions.
- Dc_qdc.odf - Quench detector control functions for authorized users.
- Dc_rs.odf - Reversing switch status and access to control functions.
- Dc_rsc.odf - Reversing switch control functions for authorized users.
- Dc_so.odf - Solenoid status, voltage tap readings and etc.
- Dc_lcw.odf - Low conductivity water system status

The DMACs and TISOFT routines that execute control algorithms are described and listed on the following pages.

- Function: Interlocks - first fault decoding
 Type: DMACS script
 Trigger: "First Fault?" button on "Interlocks" picture.
 Action: Decodes output of (3) first fault modules into text description of first interlock to open.
 Affects: FF D4,2,1 (DC_FF4-1); FF Flag A,B,C

The NIM interlock modules have their first fault outputs wire-or'd to deliver a binary encoded number (FF D4-1) and three flags (FF A,B,C) to the PLC. Only one flag will occur at a time. Depending on which flag occurs, this DMACS script decodes the binary encoded number to produce a text string which identifies the first interlock to open. The output is displayed on the operator's console in response to pressing a button located in the "Interlocks" picture. The table below shows the DMACS digital input point tagnames for the hardware contacts that connect from the NIM Interlocks module to the PLC. In the table, CL = closed contact = 1; and OP = open contact = 0.

Following the table is the DMACS script that does the decoding. The combination of FF D4, FF D2 and FF D1 is defined in the database as a multistate digital input with DMACS tagname DC_FFD4-1. It is read as a decimal "value". The flags are defined in the database as simple digital inputs DC_DI4, DC_DI5 and DC_DI6. They are read as ascii values of CLOSE or OPEN. The DMACS script is located in the "First Fault?" button in the "Interlocks" picture DC_IN.ODF.

FF D4 (DC_DI1)	FF D2 (DC_DI2)	FF D1 (DC_DI3)	value	First Fault If FF A Flag = CL (DC_DI4)	First Fault If FF B Flag = CL (DC_DI5)	First Fault If FF C Flag = CL (DC_DI6)
CL	CL	CL	7	Lead A	Controls Pmt	Dump Sw Pmt
CL	CL	OP	6	Transition A	Accel Pmt	Gnd Flt Pmt
CL	OP	CL	5	Chimney A	LCW Pmt	Reverse Sw Pmt
CL	OP	OP	4	Chimney B	Crash Pmt	Xducer Valid
OP	CL	CL	3	Transition B	Steel Pmt	DC Overcurrent
OP	CL	OP	2	Lead B	Flow Pmt	Access Gate Pmt
OP	OP	CL	1	CT Imbalance	Temp Pmt	Smoke Pmt
OP	OP	OP	0	QT Imbalance	Filter Pmt	Cryo Pmt

```

;Script to decode contents of first
;fault latch.  Resides in DC_IN.ODF
;picture file - Identify First
;Fault Button
;
DECLARE #FIRST NUMERIC SCRIPT
DECLARE #FLAGA STRING SCRIPT
DECLARE #FLAGB STRING SCRIPT
DECLARE #FLAGC STRING SCRIPT
;
GETVAL CCRS2:DC_FFD4-1.F_CV #FIRST
GETVAL CCRS2:DC_DI4.A_CV #FLAGA
GETVAL CCRS2:DC_DI5.A_CV #FLAGB
GETVAL CCRS2:DC_DI6.A_CV #FLAGC
;
IF #FLAGA == "CLOSE"
  GOTO GROUPA
  ENDIF
IF #FLAGB == "CLOSE"
  GOTO GROUPB
  ENDIF
IF #FLAGC == "CLOSE"
  GOTO GROUPC
  ELSE
    NOTE "First Fault Latch is Clear"
  ENDIF
GOTO END
;
&GROUPA
IF #FIRST == 7
  NOTE "lead A"
  ENDIF
IF #FIRST == 6
  NOTE "Transition Lead A"
  ENDIF
IF #FIRST == 5
  NOTE "Chimney Lead A"
  ENDIF
IF #FIRST == 4
  NOTE "Chimney Lead B"
  ENDIF
IF #FIRST == 3
  NOTE "Transition Lead B"
  ENDIF
IF #FIRST == 2
  NOTE "Lead B"
  ENDIF
IF #FIRST == 1
  NOTE "Center Tap Imbalance"
  ENDIF
IF #FIRST == 0
  NOTE "Quarter Tap Imbalance"
  ENDIF
GOTO end
;

&GROUPB
IF #FIRST == 7
  NOTE "Controls Permit"
  ENDIF
IF #FIRST == 6
  NOTE "Accelerator Permit"
  ENDIF
IF #FIRST == 5
  NOTE "LCW Permit"
  ENDIF
IF #FIRST == 4
  NOTE "Crash Button Permit"
  ENDIF
IF #FIRST == 3
  NOTE "Steel Position Permit"
  ENDIF
IF #FIRST == 2
  NOTE "Flow Permit"
  ENDIF
IF #FIRST == 1
  NOTE "Temperature Permit"
  ENDIF
IF #FIRST == 0
  NOTE "Filter Permit"
  ENDIF
GOTO END
;
&GROUPC
IF #FIRST == 7
  NOTE "Dump Switch Permit"
  ENDIF
IF #FIRST == 6
  NOTE "Ground Fault Permit"
  ENDIF
IF #FIRST == 5
  NOTE "Reversing Switch Permit"
  ENDIF
IF #FIRST == 4
  NOTE "Xducer Valid Permit"
  ENDIF
IF #FIRST == 3
  NOTE "DC Overcurrent"
  ENDIF
IF #FIRST == 2
  NOTE "Rm511 Access Gate Permit"
  ENDIF
IF #FIRST == 1
  NOTE "Smoke Permit"
  ENDIF
IF #FIRST == 0
  NOTE "Cryo Permit"
  ENDIF
;
&END

```

-
- **Function:** Power Supply Control Access - ON, OFF, RESET, CURRENT
 - Type:** DMACS script
 - Trigger:** "Control" button on "Power Supply" picture
 - Action:** Checks users security access before enabling access to power supply controls
 - Affects:** Enables or disables access to "Power Supply Control" sub-picture

A viewer may access power supply "status" with any level of privilege including PUBLIC. This is accomplished by the user selecting the "Power Supply" symbol of the "DZERO Solenoid Energization & Control" screen. However, only authorized users must be allowed to actually "control" the power supply. Once a user selects the power supply symbol, he is provided with an updating display of all the power supply status information. He is also offered a "Control" button for access to ON, OFF, RESET, and CURRENT settings. If he selects this button, then a "Control" screen replaces the "Status" screen. The "Control" screen is identical to the "Status" screen except that it also has the control buttons. When the "Control" screen first opens, this script runs (opening of DC_PSC.ODF) and requests the user to "log in", checks "access privileges" and provides the control buttons if access is authorized. If the user fails the login security check, then the "Status" screen is restored and control is not allowed.

```

;Script to test user privileges and grant or deny access to this
;control page
LOGIN
DECLARE #RIGHTS NUMERIC
CHECKRIGHTS "D0 Sol DC" #RIGHTS
;
IF #RIGHTS == 1
  GOTO PROCEED
  ELSE
    MESSAGE "Unauthorized Access to Solenoid Power Supply Control Attempted"
    NOTE "Security Violation -- Control Denied"
ENDIF
;
  IF #SERV == "on"
    REPLACEPIC * S:\WDMACS\PIC\D0\DC_PS
  ;
  ELSE
    REPLACEPIC * C:\WDMACS\PIC\D0\DC_PS
  ;
  ENDIF
;
&PROCEED
; Define variables local to this picture for the target current
; and desired charge/discharge rate
;
  DECLARE #TARGET NUMERIC PICTURE
  DECLARE #VLIMIT NUMERIC PICTURE

; Set initial target current to present PLC target
  GETVAL D0_CCRS2:DC_TARGET.F_CV #TARGET

; Set initial voltage limit to present PLC voltage limit
  GETVAL D0_CCRS2:DC_VLIMIT.F_CV #VLIMIT

```

- **Function:** Turn On (Power Supply)
Trigger: "Turn On" button on "Power Supply Control" window Dc_psc.odf
Action: Signals PLC to turn on power supply
Affects: D0_CCRS2:DC_CR2.F_CV

The user will have qualified to access this command by logging in to the "Power Supply Commands" screen described above. Once here, no further checks are made. By selecting the "Turn On" button, the following DMACS script will run to signal the PLC to energize the PS ON relay to the end rack for 1.0 Second. A message is sent to the alarm area file when the power supply is turned on. All interlocking is done in hardware. If the interlocks are not valid, cycling the PS ON relay will have no effect on the power supply.

```
;Turns on power supply as follows
;Sets internal PLC coil D0_CR2 closed which signals PLC
;to energize actual PS ON relay DC_RO3 for 1.0 seconds
CLOSEDIG D0_CCRS2:DC_cr2.f_CV
MESSAGE "Power supply turned on via DMACS"
```

- **Function:** Turn Off (Power Supply)
Trigger: "Turn Off" button on "Power Supply Control" picture Dc_psc.odf
Action: Turns OFF power supply
Affects: D0_CCRS2:DC_CR4.F_CV

The user will have qualified to access this command by logging in to the "Power Supply Commands" screen described above. Once here, no further checks are made. By selecting a "Turn Off" button, the following DMACS script will run to signal the PLC to momentarily energize the PS OFF relay to the end rack for 1.0 Second. The power supply OFF input is actually wired to the NC contact, which opens when energized. A message is sent to the alarm area file when the power supply is turned off.

```
;Turns off power supply as follows
;Close internal PLC contact CR4 which triggers a 1.0 second
;momentary energizing of PS OFF control relay DC_RO4
closedig D0_CCRS2:DC_CR4.F_CV
message "Power supply turned off via DMACS"
```

- **Function:** Power Supply RESET
Trigger: "Reset" button on "Power Supply Control" picture Dc_psc.odf
Action: Signals PLC to RESET power supply
Affects: D0_CCRS2:DC_CR6.F_CV

The user will have qualified to have access to this command by logging in to the "Power Supply Commands" screen described above. Once here, no further checks are made. By selecting the "Reset" button, the following DMACS script will run to signal the PLC to momentarily energize the PS RES relay to the end rack for 1.0 Seconds. A message is sent to the alarm area file when the power supply is reset.

```
;RESETs power supply as follows
;Closes internal PLC coil CR6 which triggers a momentary
;energization of PS RESET relay DC_RO5 for 1.0 seconds
closedig D0_CCRS2:DC_CR6.F_CV
message "Power supply RESET via DMACS"
```

-
- **Function:** Interlock Control Access - Reset Interlocks, Permit On, Permit Off
Trigger: "Control" button on "Interlocks" picture
Action: Checks users security access before enabling Interlock controls
Affects: Enables or disables access to "Interlock Control" sub-picture

A viewer may access interlock "status" with any level of privilege including PUBLIC. This is accomplished by the user selecting the "Interlock" symbol of the "DZERO Solenoid Energization & Control" screen. However, only authorized users must be allowed to actually reset or "control" those interlocks which are software controllable. Once a user selects the interlock symbol, he is provided with an updating display of all the interlock status information. He is also offered a "Control" button for access to Reset Interlocks, Permit On and Permit Off. If he selects this button, then a "Control" screen replaces the "Status" screen. The "Control" screen is identical to the "Status" screen except that it also has the control buttons. When the "Control" screen first opens, this script runs (opening of Dc_inc.odf) and requests the user to "log in", checks "access privileges" and provides the control buttons if access is authorized. If the user fails the login security check, then the "Status" screen is restored and control is not allowed.

```
;Script to test user privileges and grant or deny access to this
;control page.
LOGIN
DECLARE #RIGHTS NUMERIC
CHECKRIGHTS "D0 Sol DC" #RIGHTS
IF #RIGHTS == 1
  GOTO END
  ELSE
    MESSAGE "Unauthorized Access to Interlock Controls Attempted"
    NOTE "Security Violation -- Control Denied"
;   IF #SERV == "on"
      REPLACEPIC * S:\WDMACS\PIC\D0\DC_IN
;   ELSE
;   REPLACEPIC * C:\WDMACS\PIC\D0\DC_IN
;   ENDIF
ENDIF
&END
```

-
- **Function:** Enable Controls Permit
Trigger: "Permit On" button on the "Interlocks Control" picture Dc_inc.odf
Action: Signals PLC to Enable "Controls Permit" to Interlocks Module
Affects: D0_CCRS2:DC_CR10.F_CV

The user will have qualified to have access to this command by logging in to the "Interlock Control" picture described above. Once here, no further security checks are made. The Controls permit (Ctrl Pmt) is an interlock input to the interlock module. Unless this contact is closed, the power supply cannot be energized. This permit is implemented with a normally open relay contact in the control system and may be controlled by other software as well as this operator control button. This script runs when the operator selects "Permit On" while in the "Interlocks Control" screen. A message is sent to the alarm area file and to the operator when the control permit is enabled.

```
; Signals PLC to Enable the Controls Permit to the Interlock chassis
CLOSEDIG D0_CCRS2:DC_CR10.F_CV
MESSAGE "Ctrl Pmt enabled from console"
NOTE "Control Permit Enabled"
```

-
- Function: Disable Controls Permit
Trigger: "Permit Off" button on "Interlock Control" picture Dc_inc.odf
Action: Signals PLC to Disable Controls Permit to the Interlocks Module
Affects: D0_CCRS2:DC_CR10.F_CV

The user will have qualified to have access to this command by logging in to the "Interlock Control" picture described above. Once here, no further checks are made. The Controls permit (Ctrl Pmt) is an interlock to the interlock module. Unless this contact is closed, the power supply cannot be energized. This permit is implemented with a normally open relay contact in the control system and may be controlled by other software as well as this operator control button. This script runs when the operator selects "Permit Off" while in the "Interlock Control" screen. A message is sent to the alarm file and to the operator when the control permit is removed.

```
; Disables the Controls Permit to the Interlock chassis
OPENDIG D0_CCRS2:DC_CR10.F_CV
MESSAGE "Ctrl Pmt disabled from console"
NOTE "Control Permit removed"
```

-
- Function: Emergency Shutoff
Trigger: "Emergency Shutoff" button on top level picture Dcontrol.odf
Affects: D0_CCRS2:DC_CR10.F_CV
Action: Signals PLC to Disable Controls Permit to the Interlocks Module

The Controls permit (Ctrl Pmt) is an interlock to the interlock module. Unless this contact is closed, the power supply cannot be energized. This permit is implemented with a normally open relay contact in the control system and may be controlled by other software as well as this operator control button. A script runs when the operator selects "Emergency Shutoff" on the "DC Controls" top level screen. The script only opens a large, bright red picture Dc_emr.odf for confirmation by the user before shutting down. This picture provides two buttons: "Proceed" and "Cancel". If "Cancel" is selected, then the picture is closed with no further action. However, if "Proceed" is selected, then the following script is executed. Due to the emergency nature of this control, the script does not establish any authority for the operation before executing. Completion of emergency shutoff is reported to the operator and the alarm file.

```
; Emergency Shutoff
; Disables the Controls Permit to the Interlock chassis
OPENDIG D0_CCRS2:DC_CR10.F_CV
MESSAGE "Emergency Shutoff from console"
NOTE "Emergency Shutoff Commanded to PLC"
CLOSEPIC VIEWPORT
```

** This note is in progress -- In fact, all routines are finished and documented and have yet to be added to this note as of 3/16/98. Additionally, PLC ladder logic may be added.